

CALCUL DES VARIATIONS

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Le calcul des variations est automatisé dans une large mesure dans le paquetage dédié **VariationalCalculus** dont ce chapitre explore les deux commandes majeures pour l'économiste puis donne des exemples d'application en mathématiques et en économie.

Présentation du paquetage **VariationalCalculus**

Maple fournit dans sa bibliothèque un "ouvrage" consacré au calcul des variations dont l'objet est de déterminer les extrémales de fonctionnelles qui s'interprètent comme des objectifs (ou cibles) à atteindre. Ce problème n'est pas simple puisque l'inconnue ou l'ensemble des inconnues est une fonction ou un ensemble de fonctions, une trajectoire ou un ensemble de trajectoires en dynamique, qui est déterminée en définitive par la résolution d'équations différentielles.

Le paquetage dédié **VariationalCalculus** est relativement pauvre en commandes puisqu'il ne propose que cinq commandes : **ConjugateEquation**, **Convex**, **EulerLagrange**, **Jacobi** et **Weierstrass**. **EulerLagrange** est la plus importante en économie et gestion. **Convex** apporte des renseignements précieux sur la nature d'une trajectoire candidate.

La commande **EulerLagrange**

La commande **EulerLagrange** détermine, quand c'est possible, les fonctions remplissant les conditions **nécessaires** d'optimalité, qu'on appelle les conditions d'Euler-Lagrange (ou Euler tout court).

Considérons le cas le plus simple. Soit J la fonctionnelle du type $J = \int_a^b f(x(t), x'(t), t) dt$ avec $x(a) = x_a$ et $x(b) = x_b$. On cherche $x(t)$ satisfaisant les conditions aux bornes et telle que J soit extrême (optimale).

La commande **EulerLagrange** (**f(x(t), diff(x(t), t), t), t, x(t)**) où le premier argument est l'intégrande écrit sous la forme d'une **expression**, le second argument précise la variable indépendante, ici **t**, et le troisième donne la fonction inconnue, retourne un ensemble (set) renfermant selon les cas une ou deux informations :

1°) L'équation différentielle d'Euler-Lagrange, plus exactement une expression dont il est sous-entendu qu'elle est égale à 0. Maple calcule donc $\frac{d}{dx} f(x, x', t) - \frac{d}{dt} \left(\frac{d}{dx'} f(x, x', t) \right)$.

2°) **Si possible**, Maple renvoie aussi l'intégrale première de l'équation d'Euler-Lagrange par rapport à $x(t)$. Comme les conditions aux bornes ne sont pas précisées dans la commande, une constante d'intégration figure dans le résultat. Quand Maple "sèche", aucune information n'est renvoyée.

Les deux exemples qui suivent commentent les deux types de résultats.

Comme premier d'exemple, on présente le fameux problème historique dit du brachistochrone consistant à chercher la trajectoire la plus **rapide** pour qu'un point mobile soumis à l'action de la

gravité aille d'un point de coordonnées $(a, y(a))$ à un point de coordonnées $(b, y(b))$ dans un plan vertical. Formellement, le problème est de déterminer la fonction $y(x)$ qui minimise la

durée du déplacement, soit $J = \int_a^b \frac{\sqrt{1 + \left(\frac{d}{dt} y(x)\right)^2}}{y(x)} dt$ avec $y(a) = y_a$ et $y(b) = y_b$.

> restart;

with(VariationalCalculus):#chargement du paquetage

f := (1 + diff(y(x),x)^2)^(1/2)/y(x)^(1/2);# écriture de l'intégrande sous forme d'expression

sol:=EulerLagrange(f,x,y(x));

$$f := \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{\sqrt{y(x)}}$$

$$\text{sol} := \left\{ -\frac{1}{2} \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{y(x)^{3/2}} + \frac{\left(\frac{d}{dx} y(x)\right)^2 \left(\frac{d^2}{dx^2} y(x)\right)}{\left(1 + \left(\frac{d}{dx} y(x)\right)^2\right)^{3/2} \sqrt{y(x)}} + \frac{1}{2} \frac{\left(\frac{d}{dx} y(x)\right)^2}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} y(x)^{3/2}} - \frac{\frac{d^2}{dx^2} y(x)}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} \sqrt{y(x)}}, \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{\sqrt{y(x)}} - \frac{\left(\frac{d}{dx} y(x)\right)^2}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} \sqrt{y(x)}} = K_1 \right\}$$

L'output renvoie un ensemble à deux éléments. Isolons les informations, à savoir l'équation d'Euler-Lagrange toujours présentée comme une expression implicitement égale à 0 et l'intégrale première. Puisque l'output est un ensemble, leur ordre d'affichage est aléatoire suivant la session ou l'ordinateur. Il faut donc convertir le résultat en une liste puis assigner clairement l'équation d'Euler-Lagrange et l'intégrale première. Le petit programme suivant exécute ce tri.

> solist:=convert(sol,list):#conversion de l'ensemble en une liste

if type(solist[1])='=' then eulag,inteulag:=solist[2], solist[1]

else eulag,inteulag:=solist[1],solist[2]

end if;

$$\begin{aligned}
\text{eulag, inteulag} := & -\frac{1}{2} \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{y(x)^{3/2}} + \frac{\left(\frac{d}{dx} y(x)\right)^2 \left(\frac{d^2}{dx^2} y(x)\right)}{\left(1 + \left(\frac{d}{dx} y(x)\right)^2\right)^{3/2} \sqrt{y(x)}} \\
& + \frac{1}{2} \frac{\left(\frac{d}{dx} y(x)\right)^2}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} y(x)^{3/2}} - \frac{\frac{d^2}{dx^2} y(x)}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} \sqrt{y(x)}}, \\
& \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{\sqrt{y(x)}} - \frac{\left(\frac{d}{dx} y(x)\right)^2}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} \sqrt{y(x)}} = K_1
\end{aligned}
\tag{1.1.1}$$

On peut alors demander la résolution de l'équation d'Euler-Lagrange **eulag** avec **dsolve** même si on sait que l'équation différentielle est non linéaire et d'ordre 2.

> **sol_eulag:=dsolve(eulag=0,y(x));**

$$\begin{aligned}
\text{sol_eulag} := & -\sqrt{-y(x)^2 + y(x) _C1} + \frac{1}{2} _C1 \arctan\left(\frac{y(x) - \frac{1}{2} _C1}{\sqrt{-y(x)^2 + y(x) _C1}}\right) - x \\
& - _C2 = 0, \sqrt{-y(x)^2 + y(x) _C1} - \frac{1}{2} _C1 \arctan\left(\frac{y(x) - \frac{1}{2} _C1}{\sqrt{-y(x)^2 + y(x) _C1}}\right) - x \\
& - _C2 = 0
\end{aligned}$$

Maple renvoie des solutions implicites. On retrouve un résultat équivalent en résolvant l'équation différentielle du premier ordre **inteulag**.

> **dsolve(inteulag,y(x));**

$$\begin{aligned}
x + \frac{\sqrt{-K_1^2 y(x)^2 + y(x)}}{K_1} - \frac{1}{2} \frac{\arctan\left(\frac{\sqrt{K_1^2} \left(y(x) - \frac{1}{2 K_1^2}\right)}{\sqrt{-K_1^2 y(x)^2 + y(x)}}\right)}{K_1 \sqrt{K_1^2}} - _C1 = 0, x \\
- \frac{\sqrt{-K_1^2 y(x)^2 + y(x)}}{K_1} + \frac{1}{2} \frac{\arctan\left(\frac{\sqrt{K_1^2} \left(y(x) - \frac{1}{2 K_1^2}\right)}{\sqrt{-K_1^2 y(x)^2 + y(x)}}\right)}{K_1 \sqrt{K_1^2}} - _C1 = 0
\end{aligned}$$

Passons à un second exemple, économique celui-là. Il s'agit de trouver la trajectoire de consommation $c(t)$ qui maximise sur la période $[0, 10]$ les flux d'utilité escomptée au taux ρ . La consommation instantanée est la différence entre le revenu du capital $K(t)$ placé au taux d'intérêt constant r et sa dépréciation $K'(t)$: $C(t) = rK(t) - K'(t)$.

La fonction d'utilité instantanée est une fonction puissance : $U(c) = c^\alpha$, avec $0 < \alpha < 1$.
 Le consommateur démarre avec le capital $K(0) = 0$ et finit avec le capital $K(10) = 5$.
 Il s'agit en définitive de rechercher la trajectoire de $K(t)$ qui maximise

$$J = \int_0^{10} \exp(-\rho t) (rK(t) - K'(t))^\alpha dt \text{ avec les conditions } K(0) = 0 \text{ et } K(10) = 5.$$

```
> g := exp(-rho*t)*(r*K(t)-diff(K(t),t))^(1/2);# écriture de
l'intégrande
eulag:=EulerLagrange(g,t,K(t));
```

$$g := e^{-\rho t} \sqrt{rK(t) - \left(\frac{d}{dt} K(t)\right)}$$

$$eulag := \left\{ \begin{aligned} & \frac{1}{2} \frac{e^{-\rho t} r}{\sqrt{rK(t) - \left(\frac{d}{dt} K(t)\right)}} - \frac{1}{2} \frac{\rho e^{-\rho t}}{\sqrt{rK(t) - \left(\frac{d}{dt} K(t)\right)}} \\ & - \frac{1}{4} \frac{e^{-\rho t} \left(r \left(\frac{d}{dt} K(t) \right) - \left(\frac{d^2}{dt^2} K(t) \right) \right)}{\left(rK(t) - \left(\frac{d}{dt} K(t) \right) \right)^{3/2}} \end{aligned} \right\} \quad (1.1.2)$$

L'output est maintenant un ensemble à un seul élément, qui est forcément l'équation d'Euler-Lagrange. L'intégrale première n'a pas été calculée parce que l'équation est mal reconnue par

dsolve. Réécrivons-la sous sa forme résolue en $\frac{d^2}{dt^2} K(t)$.

```
> eulag:=isolate(eulag[1],diff(K(t),t,t));
```

$$eulag := \frac{d^2}{dt^2} K(t) = \frac{1}{e^{-\rho t}} \left(4 \left(-\frac{1}{2} \frac{e^{-\rho t} r}{\sqrt{rK(t) - \left(\frac{d}{dt} K(t)\right)}} + \frac{1}{2} \frac{\rho e^{-\rho t}}{\sqrt{rK(t) - \left(\frac{d}{dt} K(t)\right)}} \right) \left(rK(t) - \left(\frac{d}{dt} K(t)\right) \right)^{3/2} + r \left(\frac{d}{dt} K(t)\right) \right) \quad (1.1.3)$$

La résolution est maintenant possible et on en profite pour préciser les conditions aux bornes :

```
> sol:=simplify(dsolve({eulag} union {K(0)=0,K(10)=5},K(t)));
```

$$sol := K(t) = -\frac{5(e^{rt} - e^{2(r-\rho)t})}{-e^{10r} + e^{20r-20\rho}} \quad (1.1.4)$$

On en tire la trajectoire-solution en tant que fonction du temps.

```
> assign(sol);
```

```
F:=unapply(K(t),t);#trajectoire-solution
```


$$F := t \rightarrow -\frac{5(e^{rt} - e^{2(r-\rho)t})}{-e^{10r} + e^{20r-20\rho}} \quad (1.1.5)$$

Une représentation graphique de la trajectoire est possible à condition de préciser les valeurs des

paramètres.

```
> r,rho:=0.04,0.025;#assignation de valeurs aux paramètres  
plot(F(t),t=0..10,view=[0..12,0..6]);#représentation  
graphique
```

$r, \rho := 0.04, 0.025$



La commande **Convex**

La commande **Convex** permet de conclure sur la nature de l'extrémale, c'est à dire la solution de l'équation d'Euler-Lagrange. En effet, si l'intégrande $f(x(t), x'(t), t)$ est convexe par rapport à x et x' , alors l'extrémale correspond à un **minimum global** de la fonctionnelle.

La syntaxe est la suivante : **Convex(f(x(t),diff(x(t),t),t),t,x(t))** où le premier argument est l'intégrande, le second la variable indépendante, le troisième est la fonction inconnue.

La réponse renvoyée en écho dépend de la nature de l'intégrande. Si l'intégrande est convexe, alors Maple retourne une suite (sequence) de deux éléments : le premier donne l'expression complète de la matrice hessienne de la fonction f par rapport à x et x' ; le second est une expression logique (par exemple une inégalité) qui est vraie si et seulement si la matrice hessienne est semi-définie positive. Par exemple :

> **Convex(f,x,y(x));**

$$\frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{\sqrt{y(x)}}$$

$$\left[\left[\frac{3}{4} \frac{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2}}{y(x)^{5/2}}, -\frac{1}{2} \frac{\frac{d}{dx} y(x)}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} y(x)^{3/2}}, \right. \right.$$

$$\left. \left[-\frac{1}{2} \frac{\frac{d}{dx} y(x)}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} y(x)^{3/2}}, -\frac{\left(\frac{d}{dx} y(x)\right)^2}{\left(1 + \left(\frac{d}{dx} y(x)\right)^2\right)^{3/2} \sqrt{y(x)}} \right. \right.$$

$$\left. \left. + \frac{1}{\sqrt{1 + \left(\frac{d}{dx} y(x)\right)^2} \sqrt{y(x)}} \right] \right], 0 \leq -\frac{1}{4} \frac{\left(\frac{d}{dx} y(x)\right)^2 - 3}{\left(1 + \left(\frac{d}{dx} y(x)\right)^2\right) y(x)^3} \text{ and } 0$$

$$\leq \frac{1}{4} \frac{4 y(x)^2 + 3 + 6 \left(\frac{d}{dx} y(x)\right)^2 + 3 \left(\frac{d}{dx} y(x)\right)^4}{\left(1 + \left(\frac{d}{dx} y(x)\right)^2\right)^{3/2} y(x)^{5/2}}$$

Si l'intégrande n'est pas convexe, Maple renvoie la réponse laconique **false**.

Enfin, si Maple est indécis sur la nature de l'intégrande, il retourne la matrice hessienne et un message signalant que le recours à la procédure **IsDefinite** du paquetage

LinearAlgebra n'a pas permis de conclure.

Bien entendu, la commande **Convex** peut informer sur la concavité de l'intégrande. Il suffit de passer en argument l'opposé de la fonction sous le signe somme.

Exemples mathématiques

Les exemples de cette section sont volontairement simples. On cherche les extrêmes d'une intégrale et on étudie leur nature (maximum ou minimum). Ils se traitent de manière mécanique.

Après avoir appelé le paquetage **VariationalCalculus**, on assigne l'intégrande, qu'on passe en argument de **EulerLagrange** et de **Convex**. Pour déterminer explicitement les extrêmes, on a recours à **dsolve** appliquée à l'équation d'Euler-Lagrange en tenant compte des conditions aux bornes.

Exemple 1

Cherchons les extrêmes et leur nature du programme standard ext

$$J = \int_0^2 (12 t x(t) - 2 x'(t)) dt \text{ avec } x(0) = 0 \text{ et } x(2) = 8.$$

Les préparatifs sont rapides : appel au paquetage **VariationalCalculus** et écriture de l'intégrande.

```
> restart;
with(VariationalCalculus):
f:=12*t*x(t)+diff(x(t),t)^2;
```

$$f := 12 t x(t) + \left(\frac{d}{dt} x(t) \right)^2 \quad (2.1.1)$$

L'appel à la commande **EulerLagrange** amène ici l'équation d'Euler-Lagrange et elle seule. Curieusement, Maple ne renvoie pas son intégrale première (trop facile?).

```
> sol:=EulerLagrange(f,t,x(t));
eulag:=sol[1];
```

$$\text{sol} := \left\{ 12 t - 2 \left(\frac{d^2}{dt^2} x(t) \right) \right\}$$

$$\text{eulag} := 12 t - 2 \left(\frac{d^2}{dt^2} x(t) \right) \quad (2.1.2)$$

On détermine la solution tenant compte des conditions aux bornes.

```
> sol_eulag:=dsolve({eulag,x(0)=0,x(2)=8},x(t));
sol_eulag:=x(t)=t^3
```

$$\text{sol_eulag} := x(t) = t^3 \quad (2.1.3)$$

La nature de l'extrémale est précisée par **Convex**.

```
> Convex(f,t,x(t));
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 2 \end{bmatrix}, \text{true} \quad (2.1.4)$$

L'intégrande est convexe en x et x' . Par conséquent la fonction $f(t) = t^3$ correspond à un minimum global de la fonctionnelle J .

Exemple 2

Cherchons les extrémales et leur nature du programme $\text{ext } J = \int_1^5 \left(3t + x'(t)^{\frac{1}{2}} \right) dt$ avec

$x(1) = 3$ et $x(5) = 7$.

Dans cet exemple, l'instruction **EulerLagrange** renvoie un ensemble formé de l'équation d'Euler-Lagrange et de son intégrale première. Il faut isoler l'équation d'Euler pour la traiter ultérieurement avec **dsolve**.

```
> g:=3*t+diff(x(t),t)^(1/2);
sol:=EulerLagrange(g,t,x(t));
solist:=convert(sol,list);
if type(solist[1])='=' then eulag,inteulag:=solist[2],
solist[1]
else eulag,inteulag:=solist[1],solist[2]
end if;
```

$$g := 3t + \sqrt{\frac{d}{dt} x(t)}$$

$$sol := \left\{ \frac{1}{4} \frac{\frac{d^2}{dt^2} x(t)}{\left(\frac{d}{dt} x(t)\right)^{3/2}}, \frac{1}{2 \sqrt{\frac{d}{dt} x(t)}} = K_1 \right\}$$

$$solist := \left[\frac{1}{4} \frac{\frac{d^2}{dt^2} x(t)}{\left(\frac{d}{dt} x(t)\right)^{3/2}}, \frac{1}{2 \sqrt{\frac{d}{dt} x(t)}} = K_1 \right]$$

$$eulag, inteulag := \frac{1}{4} \frac{\frac{d^2}{dt^2} x(t)}{\left(\frac{d}{dt} x(t)\right)^{3/2}}, \frac{1}{2 \sqrt{\frac{d}{dt} x(t)}} = K_1 \quad (2.2.1)$$

La résolution de l'équation d'Euler-Lagrange ne présente aucune difficulté.

$$\begin{aligned} &> \text{sol_eulag} := \text{dsolve}(\{\text{eulag}, x(1)=3, x(5)=7\}, x(t)); \\ &\text{sol_eulag} := x(t) = t + 2 \end{aligned} \quad (2.2.2)$$

Cette extrémale correspond à un maximum global de la fonctionnelle. En effet, l'instruction **Convex** renvoie la matrice hessienne de l'intégrande considéré comme une fonction de x et x' et, comme elle est diagonale, on constate que ses valeurs propres sont négatives ou nulles sur le domaine de validité du problème posé.

$$\begin{aligned} &> \text{Convex}(g, t, x(t)); \\ &\left[\begin{array}{cc} 0 & 0 \\ 0 & -\frac{1}{4 \left(\frac{d}{dt} x(t)\right)^{3/2}} \end{array} \right], 0 \leq -\frac{1}{4 \left(\frac{d}{dt} x(t)\right)^{3/2}} \end{aligned} \quad (2.2.3)$$

Application économique : le modèle néo-classique de croissance

En 1928, un jeune anglais, Frank Ramsey, a publié un article court, mais célèbre, dans la revue de J. M Keynes, *The Economic Review*, intitulé "A Mathematical Theory of Saving". Du point de vue formel, c'est l'une des toutes premières utilisations du calcul des variations en macroéconomie. De nos jours, il est d'usage d'y voir le prototype d'une endogénéisation du taux d'épargne national dans un modèle de croissance. Au reste, quitte à dénaturer la substance du travail de Ramsey, les macroéconomistes contemporains s'y réfèrent comme le modèle générique de la théorie néo-classique de la croissance optimale.

L'output Q est produit à partir de la combinaison productive de deux inputs, le capital K et le travail L de sorte que $Q = F(K, L)$ avec F homogène de degré 1, $F'(K) > 0$, $F'(L) > 0$, $F''_{KK} < 0$ et F''_{LL}

< 0 . Dans sa forme la plus simple, le modèle suppose qu'il n'y a pas de progrès technique et que l'emploi croît au taux constant n .

L'output se partage entre consommation, C , et investissement brut, I , égal ici à l'investissement net en supposant que le capital ne se déprécie pas. Comme l'investissement net à la date t est égal à la dérivée du capital par rapport au temps, on a $Q(t) = C(t) + K'(t)$, soit encore

$$C(t) = F(K(t), L(t)) - K'(t).$$

On sait que la propriété d'homogénéité de la fonction de production permet de raisonner "par tête", autrement dit de se focaliser sur le produit par tête, le capital par tête et la consommation par tête.

On pose : $q(t) = \frac{Q(t)}{L(t)}$, $k(t) = \frac{K(t)}{L(t)}$, $q(t) = f(k(t))$ (avec $f' > 0$ et $f'' < 0$) et $c(t) = \frac{C(t)}{L(t)}$. Alors,

l'équilibre sur le marché des produits équivaut à l'égalité : $f(k(t)) = c(t) + k'(t) + n k(t)$ ou encore $c(t) = f(k(t)) - k'(t) - n k(t)$.

Le bureau de la planification ou Planificateur Central est chargé de déterminer la trajectoires de consommation par tête qui maximise la somme des satisfactions actualisées au taux $\rho > 0$ sur le long terme, en tenant compte évidemment de la contrainte d'accumulation du capital. La fonction de satisfaction U a pour argument la consommation et ses propriétés usuelles sont $U' > 0$ et $U'' < 0$, conformément au principes de l'utilité marginale positive et décroissante.

Sur la période $[0; T]$ où T est éventuellement infinie, il s'agit de résoudre le programme :

$$\max J(c(t)) = \int_0^T e^{-\rho t} U(c(t)) dt$$

sous la contrainte $c(t) = f(k(t)) - k'(t) - n k(t)$.

En escamotant la contrainte, le problème s'écrit encore :

$$\max V(k(t)) = \int_0^T e^{-\rho t} U(f(k(t)) - k'(t) - n k(t)) dt$$

mais il est bien clair que les variables c et k sont liées par la contrainte macroéconomique.

Approche dans le cas général

On commence par créer 4 fonctions "pures" : la fonction d'utilité U qui associe au niveau de consommation un indice d'utilité instantanée, la fonction de production par tête Y , une fonction du stock de capital qui dépend du temps K et la fonction de consommation déduite de la contrainte d'équilibre macroéconomique C . A quoi s'ajoute la fonction objectif - l'intégrande, qui dépend explicitement du temps par l'introduction d'un taux de préférence pour le présent et implicitement par composition des fonctions d'utilité et de production avec le capital par tête :

> restart;

$U := c \rightarrow u(c);$ #fonction d'utilité instantanée

$Y := b \rightarrow f(b);$ #Fonction de production par tête

$K := t \rightarrow k(t);$ #le capital par tête est une simple fonction du temps

$C := t \rightarrow Y(K(t)) - \text{diff}(K(t), t) - n * K(t);$ #fonction de consommation

$\text{obj} := t \rightarrow \exp(-\rho * t) * U(C(t));$ #intégrande, encore appelé fonction objectif.

$$U := c \rightarrow u(c)$$

$$Y := b \rightarrow f(b)$$

$$K := t \rightarrow k(t)$$

$$C := t \rightarrow Y(K(t)) - \left(\frac{d}{dt} K(t) \right) - n K(t)$$

$$\text{obj} := t \rightarrow e^{-\rho t} U(C(t))$$

(3.1.1)

La commande **EulerLagrange** du paquetage **VariationalCalculus** donne l'équation d'Euler-Lagrange, c'est à dire la condition nécessaire d'optimalité. Notons que Maple ne cherche

pas ici à intégrer l'équation différentielle obtenue.

```
> with(VariationalCalculus):
sol:=EulerLagrange(obj(t),t,K(t));
eulag:=sol[1];
```

$$\begin{aligned} \text{sol} := & \left\{ e^{-\rho t} D(u) \left(f(k(t)) - \left(\frac{d}{dt} k(t) \right) - n k(t) \right) (D(f)(k(t)) - n) \right. \\ & - \rho e^{-\rho t} D(u) \left(f(k(t)) - \left(\frac{d}{dt} k(t) \right) - n k(t) \right) + e^{-\rho t} D^{(2)}(u) \left(f(k(t)) \right. \\ & - \left. \left(\frac{d}{dt} k(t) \right) - n k(t) \right) \left(D(f)(k(t)) \left(\frac{d}{dt} k(t) \right) - \left(\frac{d^2}{dt^2} k(t) \right) \right. \\ & \left. \left. - n \left(\frac{d}{dt} k(t) \right) \right) \right\} \end{aligned}$$

$$\begin{aligned} \text{eulag} := & e^{-\rho t} D(u) \left(f(k(t)) - \left(\frac{d}{dt} k(t) \right) - n k(t) \right) (D(f)(k(t)) - n) & (3.1.2) \\ & - \rho e^{-\rho t} D(u) \left(f(k(t)) - \left(\frac{d}{dt} k(t) \right) - n k(t) \right) + e^{-\rho t} D^{(2)}(u) \left(f(k(t)) \right. \\ & - \left. \left(\frac{d}{dt} k(t) \right) - n k(t) \right) \left(D(f)(k(t)) \left(\frac{d}{dt} k(t) \right) - \left(\frac{d^2}{dt^2} k(t) \right) \right. \\ & \left. - n \left(\frac{d}{dt} k(t) \right) \right) \end{aligned}$$

Réécrivons l'équation d'Euler-Lagrange sous une forme plus usuelle (pour les économistes) en faisant apparaître la variable consommation du point de vue du consommateur à la date t :

```
> subs({C(t)=c(t),diff(C(t),t)=diff(c(t),t)},eulag);
eulag_1:=%=0;eulag_2:=simplify(exp(rho*t)*eulag_1);
solve(eulag_2,D(c)(t));collect(%,D(u)(c(t)));eulag_3:=diff
(c(t),t)=%;
equa_eulag:=D(D(u))(c(t))/D(u)(c(t))*eulag_3;
```

$$\begin{aligned} & e^{-\rho t} D(u)(c(t)) (D(f)(k(t)) - n) - \rho e^{-\rho t} D(u)(c(t)) \\ & + e^{-\rho t} D^{(2)}(u)(c(t)) \left(\frac{d}{dt} c(t) \right) \end{aligned}$$

$$\begin{aligned} \text{eulag}_1 := & e^{-\rho t} D(u)(c(t)) (D(f)(k(t)) - n) - \rho e^{-\rho t} D(u)(c(t)) \\ & + e^{-\rho t} D^{(2)}(u)(c(t)) \left(\frac{d}{dt} c(t) \right) = 0 \end{aligned}$$

$$\begin{aligned} \text{eulag}_2 := & D(u)(c(t)) D(f)(k(t)) - D(u)(c(t)) n - \rho D(u)(c(t)) \\ & + D^{(2)}(u)(c(t)) D(c)(t) = 0 \end{aligned}$$

$$\begin{aligned} & - \frac{D(u)(c(t)) (D(f)(k(t)) - n - \rho)}{D^{(2)}(u)(c(t))} \\ & - \frac{D(u)(c(t)) (D(f)(k(t)) - n - \rho)}{D^{(2)}(u)(c(t))} \end{aligned}$$

$$eulag_3 := \frac{d}{dt} c(t) = - \frac{D(u)(c(t)) (D(f)(k(t)) - n - \rho)}{D^{(2)}(u)(c(t))}$$

$$equa_eulag := \frac{D^{(2)}(u)(c(t)) \left(\frac{d}{dt} c(t) \right)}{D(u)(c(t))} = -D(f)(k(t)) + n + \rho \quad (3.1.3)$$

Il est facile d'en déduire la célèbre "règle de Keynes-Ramsey". Dans le cas où la fonction d'utilité instantanée est escomptée au taux de préférence pour le présent ρ , la règle d'or d'accumulation du capital est "modifiée" (modified golden rule) en ce sens que le taux de croissance de l'utilité marginale $-\frac{U''(c(t)) c'(t)}{U'(c(t))}$ est égal à la productivité marginale nette du capital

$f'(k(t)) - n$ diminuée du taux de préférence pour le présent ρ .

L'équation **eulag_3** montre nettement que la stratégie intertemporelle optimale de consommation dépend du capital par tête, dont les variations dépendent elles de la consommation par tête par la contrainte macroéconomique. On est conduit à étudier un système différentiel en (c, k) .

```
> equ_c:=eulag_3;
equ_k:=diff(k(t),t)=solve(c(t)=C(t),diff(k(t),t));
sys_dif:={equ_c,equ_k};
var:={c(t),k(t)};
```

$$equ_c := \frac{d}{dt} c(t) = - \frac{D(u)(c(t)) (D(f)(k(t)) - n - \rho)}{D^{(2)}(u)(c(t))}$$

$$equ_k := \frac{d}{dt} k(t) = -c(t) + f(k(t)) - n k(t)$$

$$sys_dif := \left\{ \frac{d}{dt} c(t) = - \frac{D(u)(c(t)) (D(f)(k(t)) - n - \rho)}{D^{(2)}(u)(c(t))}, \frac{d}{dt} k(t) = -c(t) + f(k(t)) - n k(t) \right\}$$

$$var := \{c(t), k(t)\} \quad (3.1.4)$$

A ce niveau de généralité, il est vain de demander une résolution explicite ou qualitative par Maple. Il est grand temps de spécifier les fonctions de consommation et de production pour visualiser des orbites caractérisant la dialectique de l'accumulation du capital et de la consommation optimale.

Approche avec spécification des fonctions d'utilité et de production

On retient une fonction d'utilité à élasticité intertemporelle de substitution constante (CIES : Constant Intertemporal Elasticity of Substitution) et une fonction de production Cobb-Douglas.

```
> u:=c->c^(1-theta)/(1-theta);#fonction d'utilité CIES
f:=b->b^alpha;#fonction de production Cobb-Douglas
sys_spe:=simplify(sys_dif);
```

$$u := c \rightarrow \frac{c^{1-\theta}}{1-\theta}$$

$$f := b \rightarrow b^\alpha$$

$$\text{sys_spe} := \left\{ \frac{d}{dt} c(t) = \frac{c(t) (k(t)^{\alpha-1} \alpha - n - \rho)}{\theta}, \frac{d}{dt} k(t) = -c(t) + k(t)^\alpha - n k(t) \right\} \quad (3.2.1)$$

Vue la non linéarité du système différentiel, il est inutile de demander une résolution explicite. Passons immédiatement à l'étude qualitative en spécifiant les paramètres.

```
> sys_num:=subs({alpha=1/3,theta=1/2,n=0.02,rho=0.01},
  sys_spe);
```


$$\text{sys_num} := \left\{ \frac{d}{dt} c(t) = 2 c(t) \left(\frac{1}{3 k(t)^{2/3}} - 0.03 \right), \frac{d}{dt} k(t) = -c(t) + k(t)^{1/3} - 0.02 k(t) \right\} \quad (3.2.2)$$

Avec ces valeurs, on est en mesure d'explorer rapidement le problème à l'aide du paquetage **DEtools** avant de s'engager dans une étude "standard", plus proche des manuels économiques.

Approche qualitative Maple de la dynamique

Les champs directionnels sont directement donnés par la commande **dfieldplot** du paquetage **DEtools**.

```
> with(DEtools):
  graf1:=dfieldplot(sys_num,[k(t),c(t)],t=0..100,k=0..125,c=
  0..6):graf1;
```




Le graphique met bien en évidence la propriété de point-selle de l'équilibre. Elle est encore plus visible par la commande **DEplot** qui permet d'inclure quelques orbites caractéristiques en précisant leurs conditions initiales.

```
> graf2:=DEplot(sys_num,[k(t),c(t)],t=0..100,[[k(0)=10,c(0)=1],[k(0)=20,c(0)=0.5],[k(0)=120,c(0)=6],[k(0)=90,c(0)=6]],k=0..125,c=0..6,linecolor=black):graf2;
```

Warning, plot may be incomplete, the following errors(s) were issued:

```
cannot evaluate the solution further right of 33.826376, probably a singularity
```



Approche qualitative standard de la dynamique

La méthode standard suit quatre étapes : repérage des équilibres dynamiques; détermination des isoclines nuls; étude de la stabilité des équilibres; construction du diagramme des phases.

Equilibre stationnaire

Rappelons qu'un équilibre dynamique est une position invariante dans le temps. On résout donc le système différentiel sous la condition $\frac{d}{dt} k(t) = \frac{d}{dt} c(t) = 0$.

```
> sys_stat:=subs({diff(k(t),t)=0,diff(c(t),t)=0},sys_num);
etat_stat:=solve(sys_stat,var);
```

$$\text{sys_stat} := \left\{ 0 = 2 c(t) \left(\frac{1}{3 k(t)^{2/3}} - 0.03 \right), 0 = -c(t) + k(t)^{1/3} - 0.02 k(t) \right\}$$

```
etat_stat := {c(t) = 1. 10-13, k(t) = 353.5533906}, {c(t) = 2.592592593, k(t)
= 37.03703704}
```

Bien entendu, on retient la solution "intéressante" pour laquelle la consommation de long terme n'est pas nulle.

```
> c_stat:=2.592592593;#consommation à l'état stationnaire
```

```

k_stat:=37.03703704;#capital par tête à l'état
stationnaire
eq_stat:=[k_stat,c_stat];#équilibre stationnaire
c_stat := 2.592592593
k_stat := 37.03703704
eq_stat := [37.03703704, 2.592592593]

```

Isoclines

L'isocline $\frac{d}{dt} c(t) = 0$ a pour équation : $2 \left(\frac{1}{3 k(t)^{\frac{2}{3}}} - 0.03 \right) c(t) = 0$. On résout cette dernière par rapport à $k(t)$ puisque le cas d'une consommation perpétuellement nulle n'offre guère d'intérêt.

```

> iso_c:=solve(rhs(sys_num[1])=0,k(t));
iso_c := 37.03703704

```

(3.4.2.1)

L'isocline $\frac{d}{dt} k(t) = 0$ a pour équation : $0 = k(t)^{\frac{1}{3}} - c(t) - 0.02 k(t)$ et se résout par rapport à $c(t)$.

```

> iso_k:=solve(rhs(sys_num[2])=0,c(t));
iso_k := k(t)^{1/3} - 0.02000000000 k(t)


```

Dans le plan Okc , on a la représentation graphique suivante :

```

> graf3:=plot([iso_c,t,t=0..10],k=0..355,c=0..6,color=
black,thickness=2):
graf4:=plot(iso_k,k=0..355,c=0..6,color=black,thickness=
2):
with(plots):
graf5:=display(graf3,graf4):graf5;


```



L'équilibre dynamique ne passe pas par le niveau de consommation maximale. Cette propriété est tout à fait fondamentale. Dès qu'on introduit un taux de préférence pour le présent dans un modèle de croissance néo-classique, la consommation de long terme n'est pas celle déduite de la règle d'or. Potentiellement, l'économie est en sous-accumulation du capital et cette position est efficiente.

Comme on le fait usuellement dans les manuels, on veut représenter dans le même graphique les isoclines et la droite verticale représentant le capital par tête de la règle d'or. Ce dernier s'obtient en détectant argmax de la courbe `iso_k`. Après quoi, on superpose les graphiques dans le but de montrer nettement la déviation vers la gauche du capital par tête en présence d'une préférence pour le présent des agents consommateurs.

```
> isok:=subs(k(t)=k,iso_k);#isocline k'(t)=0
k_goldenrule:=solve(diff(isok,k)=0,k);#annulation de la
dérivée
graf6:=plot([k_goldenrule,t,t=0..10],k=0..355,c=0..6,
color=blue,thickness=2):
graf7:=display(graf5,graf6):graf7;
isok := k1/3 - 0.020000000000 k
k_goldenrule := 68.04138174
```

▼ *Propriétés locales de l'équilibre et détermination du bras stable et du bras instable*

Les propriétés locales de l'équilibre peuvent être mises en évidence en linéarisant le système différentiel au voisinage de l'équilibre puis en examinant les caractéristiques de la matrice jacobienne.

Commençons par linéariser le système différentiel en (k, c) . Puisque Maple refuse de dériver les membres de droite du système `sys_num` par rapport à $c(t)$ et $k(t)$, il faut procéder à quelques manipulations pour extraire la jacobienne.

```
> sys_numer:=[lhs(sys_num[2])=subs({c(t)=c,k(t)=k},rhs
  (sys_num[2])),lhs(sys_num[1])=subs({c(t)=c,k(t)=k},rhs
  (sys_num[1]))];#réorganisation et réécriture du système
différentiel
li1:=rhs(sys_numer[1]);
f1:=unapply(li1,(k,c));#création d'une fonction en (k,c)
pour le second membre de la première équation
différentielle
li2:=rhs(sys_numer[2]);
```

**f2:=unapply(li2,(k,c));#création d'une fonction en (k,c)
pour le second membre de la seconde équation
différentielle**

**MJ:=Matrix(2,2,[[diff(f1(k,c),k),diff(f1(k,c),c)],[diff
(f2(k,c),k),diff(f2(k,c),c)]]);#matrice jacobienne du
système différentiel**

$$\text{sys_numer} := \left[\frac{d}{dt} k(t) = -c + k^{1/3} - 0.02 k, \frac{d}{dt} c(t) = 2 c \left(\frac{1}{3 k^{2/3}} - 0.03 \right) \right]$$

$$li1 := -c + k^{1/3} - 0.02 k$$

$$f1 := (k, c) \rightarrow -c + k^{1/3} - 0.02 k$$

$$li2 := 2 c \left(\frac{1}{3 k^{2/3}} - 0.03 \right)$$

$$f2 := (k, c) \rightarrow 2 c \left(\frac{1}{3 k^{2/3}} - 0.03 \right)$$

$$MJ := \begin{bmatrix} \frac{1}{3 k^{2/3}} - 0.02 & -1 \\ -\frac{4}{9} \frac{c}{k^{5/3}} & \frac{2}{3 k^{2/3}} - 0.06 \end{bmatrix}$$

(3.4.3.1)

On évalue la matrice jacobienne à l'équilibre stationnaire en remplaçant k et c par leur valeur de long terme.

> MJ_eq:=subs({k=k_stat,c=c_stat},MJ);

$$MJ_eq := \begin{bmatrix} 0.010000000000 & -1 \\ -0.002800000000 & 0. \end{bmatrix}$$

(3.4.3.2)

On dispose de tous les éléments pour écrire le système linéarisé au voisinage de l'équilibre.

> v:=<k(t),c(t)>;v_stat:=<k_stat,c_stat>;

systeme:=<diff(k(t),t),diff(c(t),t)> = MJ_eq.(v-v_stat);

#système linéarisé au voisinage de l'équilibre

$$v := \begin{bmatrix} k(t) \\ c(t) \end{bmatrix}$$

$$v_stat := \begin{bmatrix} 37.03703704 \\ 2.592592593 \end{bmatrix}$$

$$\text{systeme} := \begin{bmatrix} \frac{d}{dt} k(t) \\ \frac{d}{dt} c(t) \end{bmatrix} = \begin{bmatrix} 0.010000000000 k(t) + 2.222222223 - c(t) \\ -0.002800000000 k(t) + 0.1037037037 \end{bmatrix}$$

(3.4.3.3)

En demandant les valeurs propres de la jacobienne, on a des informations sur la nature de l'équilibre.

> with(LinearAlgebra):Eigenvectors(MJ_eq,output='list');

$$\left[\left[\left[0.0581507290636732455 + 0. I, 1, \left\{ \left[\begin{array}{l} 0.998842765536730016 + 0. I \\ -0.0480950073805691894 + 0. I \end{array} \right\}, \right. \right. \right. \quad (3.4.3.4)$$

$$\left. \left. \left. -0.0481507290636732435 + 0. I, 1, \left\{ \left[\begin{array}{l} 0.998313522278985622 + 0. I \\ 0.0580526591546466075 + 0. I \end{array} \right\} \right] \right] \right]$$

Il y a deux valeurs propres réelles : une négative et une positive. On est bien dans le cadre d'une configuration de point-selle comme l'avait montré l'approche qualitative Maple. L'équilibre est instable mais il existe une orbite ramenant vers l'équilibre. Dès lors, on recherche les équations du bras stable et du bras instable à partir des solutions explicites de **systeme**.

```
> sys_sol:=dsolve({lhs(systeme)[1]=rhs(systeme)[1],lhs
(systeme)[2]=rhs(systeme)[2]},[k(t),c(t)]);#résolution
explicite du système différentiel
sys_solf:=evalf(sys_sol);
```

$$\text{sys_sol} := \left\{ c(t) = e^{\frac{1}{200} (1 + \sqrt{113}) t} _C2 + _C1 e^{\left(\frac{1}{200} - \frac{1}{200} \sqrt{113}\right) t} \right.$$

$$\left. + \frac{36296296307}{14000000000}, k(t) = \left(-\frac{25}{14} \sqrt{113} - \frac{25}{14} \right) _C2 e^{\frac{1}{200} (1 + \sqrt{113}) t} \right.$$

$$\left. + \frac{1037037037}{28000000} + \left(-\frac{25}{14} e^{-\frac{1}{200} (-1 + \sqrt{113}) t} \right. \right.$$

$$\left. + \frac{25}{14} \sqrt{113} e^{-\frac{1}{200} (-1 + \sqrt{113}) t} \right) _C1 \}$$

$$\text{sys_solf} := \{c(t) = e^{0.05815072905 t} _C2 + _C1 e^{-0.04815072905 t} + 2.592592593, k(t) = (3.4.3.5)$$

$$-20.76811753 e^{0.05815072905 t} _C2 + 37.03703704$$

$$+ 17.19668895 _C1 e^{-0.04815072905 t}\}$$

Pour obtenir l'équation du bras stable, il faut fixer la valeur de la constante $_C2$ à 0 afin de "neutraliser" les exponentielles croissantes puis exprimer la relation entre c et k :

```
> xp1:=subs({_C2=0,k(t)=k,c(t)=c},sys_solf);xp2:=solve(xp1
[2],_C1);xp3:=subs(_C1=xp2,xp1[1]);
```

$$xp1 := \{c = 2.592592593 + _C1 e^{-0.04815072905 t}, k = 37.03703704$$

$$+ 17.19668895 _C1 e^{-0.04815072905 t}\}$$

$$xp2 := \frac{4.652058325 \cdot 10^{-9} (1.2500000 \cdot 10^7 k - 4.62962963 \cdot 10^8)}{e^{-0.04815072905 t}}$$

$$xp3 := c = 0.438861887 + 0.05815072906 k \quad (3.4.3.6)$$

L'équation du bras instable s'obtient en fixant la constante d'intégration $_C1$ à 0.

```
> yp1:=subs({_C1=0,k(t)=k,c(t)=c},sys_solf);yp2:=solve(yp1
[2],_C2);yp3:=subs(_C2=yp2,yp1[1]);
```

$$yp1 := \{c = e^{0.05815072905 t} _C2 + 2.592592593, k = -20.76811753 e^{0.05815072905 t} _C2$$


+ 37.03703704}

$$yp2 := - \frac{3.852058324 \cdot 10^{-9} (1.2500000 \cdot 10^7 k - 4.62962963 \cdot 10^8)}{e^{0.05815072905 t}}$$

$$yp3 := c = -0.04815072905 k + 4.375952928$$

Pour préparer le diagramme des phases, on représente ces deux droites dans le même graphique.


```
> graf8:=plot([rhs(xp3),rhs(yp3)],k=20..50,c=1.5..3.5,  
colour=[black,black],thickness=2):graf8;
```



▼ Diagramme des phases

Le diagramme des phases superpose les graphiques **graf2**, **graf5** et **graf8** pour visualiser les champs directionnels, quatre orbites caractéristiques, les isoclines et les bras stable et instable au voisinage de l'équilibre dynamique E .

```
> display({graf2,graf5,graf8},view=[0..120,0..6]);
```



Exercices

Exercice M1

Déterminer les extrémales éventuelles des fonctionnelles suivantes :

1. $V(x) = \int_0^1 (tx + 2x^2) dt$ avec $x(0) = 1$ et $x(1) = 2$.

2. $V(x) = \int_0^1 (2xe^t + x^2 + x') dt$ avec $x(0) = 2$ et $x(1) = 2e^2 + e^{-2}$.

3. $V(x) = \int_0^1 (tx' + x^2) dt$ avec $x(0) = 1$ et $x(1) = 0$.

$$4. V(x) = \int_0^1 (x^2 + x^2 + 2x e^t) dt.$$

$$5. V(x) = \int_1^2 \left(\frac{x^2}{t^2} \right) dt \text{ avec } x(1) = 1 \text{ et } x(2) = 2.$$

$$6. V(x) = \int_0^{\frac{\pi}{2}} (x^2 - x^2) dt \text{ avec } x(0) = 1 \text{ et } x\left(\frac{\pi}{2}\right) = 0.$$

$$7. V(x) = \int_0^1 \sqrt{1 + x^2} dt.$$

$$8. V(x) = \int_0^{\frac{\pi}{2}} (x^2 - x^2) dt \text{ avec } x(0) = 0 \text{ et } x\left(\frac{\pi}{2}\right) = 1.$$

Quand une extrémale existe, précisez sa nature.

Exercice M2

Etudier le programme :

$$\text{ext } J = \int_0^T x'(t) dt \text{ avec } x(0) = \alpha \text{ et } x(T) = \beta$$

Exercice M3

Expliquez pourquoi le programme :

$$\text{ext } J = \int_0^5 (t + y(t)^2 + 3y'(t)) dt \text{ avec } y(0) = 3 \text{ et } y(5) = 3$$

n'admet pas de solution.

Exercice M4

On considère la fonctionnelle dépendant de deux variables $V(x, y) = \int_0^T (x + y + x^2 + y^2) dt$.

Après avoir compulsé l'aide en ligne, donnez le système des équations d'Euler-Lagrange puis déterminez les extrémales $x^*(t)$ et $y^*(t)$.

Exercice M5

Soit le programme $\text{ext } J = \int_0^2 (ty^2 + yy' + y''^2) dt$ avec $x(0) = 0$ et $x(2) = 8$. Après avoir compulsé l'aide en ligne, cherchez l'expression générale de l'extrémale de J . Donnez une représentation graphique de quelques extrémales particulières.

Exercice E1 (microdynamique du monopole)

Dans l'article de G.C. Evans "The Dynamics of the Monopoly" (American Mathematical

Monthly, February 1924), un monopole produit et vend un bien unique en quantité Q . Sa fonction de coût de production est quadratique : $C = \alpha Q^2 + \beta Q + \gamma$ où α , β et γ sont des paramètres strictement positifs. La quantité produite dépend du prix et des variations de prix : $Q = a - bP + hP'$, avec $a > 0$, $b > 0$ et $h \neq 0$. Les questions suivantes permettent de saisir la substance du modèle de Evans.

1. Donner l'expression du profit **Profit** en fonction du prix et de ses variations.
2. L'objectif de la firme est de déterminer la trajectoire optimale du prix $P(t)$ qui maximise le profit sur la période $[0;T]$. Ecrire le programme intertemporel de la firme.
3. Déterminer l'extrémale du programme.
4. Représenter graphiquement l'extrémale correspondant aux valeurs de paramètres $\alpha = 0.05$, $\beta = 0.1$, $a = 100$, $b = 0.25$, $h = 1$, $T = 10$ et pour les conditions aux bornes $P(0) = P(10) = 150$.

Exercice E2 (arbitrage inflation-chômage)

Cet exercice s'inspire du modèle célèbre de Dean Taylor développé dans l'article "Stopping Inflation in the Dornbusch Model : Optimal Monetary Policies with Alternate Price-Adjustment Equations" (Journal of Macroeconomics, 1989). L'idée générale est que l'inflation et le chômage sont deux fléaux qui génèrent des pertes sociales. Si un arbitrage à la Phillips existe entre les deux, que doit être la meilleure combinaison inflation-chômage dans le temps? Dans le modèle et l'exercice proposé ci-dessous, le chômage apparaît indirectement à travers la différence entre le niveau de production de plein emploi et le niveau de production courant.

On note λ la perte sociale générée par l'inflation et le chômage, Y la production courante, Y_p la production de plein emploi, P le taux d'inflation courant et Pe le taux d'inflation anticipé.

Le modèle d'arbitrage intertemporel entre l'inflation et le chômage repose sur les trois équations suivantes :

$$(1) \lambda = (Y - Y_p)^2 + \alpha P^2, \alpha > 0.$$

$$(2) P = -\beta (Y - Y_p) + Pe, \beta > 0.$$

$$(3) \frac{d}{dt} Pe = j (P - Pe), 0 < j \leq 1.$$

L'équation (1) signifie qu'à toute date la perte sociale est la somme du carré des écarts entre l'objectif souhaitable (Pe pour la production, inflation nulle pour les prix) et la valeur courante.

L'équation (2) est la relation de Phillips augmentée des anticipations. L'équation (3) fait l'hypothèse que les anticipations de prix sont adaptatives.

On veut répondre aux trois questions suivantes :

1. Montrer qu'à toute date la perte sociale est une fonction des anticipations de prix et de leurs variations.
2. Le problème des pouvoirs publics est de sélectionner la trajectoire optimale de Pe qui minimise la perte sociale sur l'intervalle temporel $[0;T]$ en tenant compte du fait que les agents ont une préférence pour le présent mesurée par le taux ρ et que, partant du taux d'inflation anticipé $Pe(0) = Pe_0 > 0$, il faut atteindre l'objectif $Pe(T) = 0$. Ecrire le programme correspondant.
3. Déterminer l'extrémale du programme.
4. Représentez graphiquement l'extrémale en prenant $\rho = 0.05$, $\alpha = 1$, $\beta = 0.5$, $j = 0.2$ et les conditions aux bornes $Pe(0) = 15$ et $Pe(40) = 0$.